
Public CurPos


Public SRowNum


Public IsForm


Public MRow$


Public TotalCol


Sub ChemFormula()





sel$ = Selection





'Initializations


'Bericht bij een punt die mogelijk verhoogd moet worden.


DotQ$ = "Moet de punt" + Chr$(13) + "OMHOOG?"


DotQ$ = DotQ$ + Chr$(13) + "(zoals bij hydraten)"


'(deel van) Bericht als een mogelijk superscript aanwezig is.


SuperQ$ = " zijn " + Chr$(13) + "SUPERSCRIPT?"


SuperQ$ = SuperQ$ + Chr$(13) + "(Is dit onderdeel van een lading?)"


'Bericht als er samengevoegde rijen zijn.


MRow$ = "ChemFormula kan niet verder met het opmaken van deze selectie na "


MRow$ = MRow$ + "een cel met samengevoegde rijen." + Chr$(13)


MRow$ = MRow$ + "Probeer ChemFormula op een kleiner deel van de selectie."


'Bericht als er niets geselecteerd is.


NoText$ = "Selecteer de op te maken tekst"


'TableString$ is aanwezig in tabellen,  wordt gebruikt om ze te identificeren


TableString$ = Chr$(13) + Chr$(7)


'IsForm geeft aan of huidige selectie deel van een formule is


IsForm = 0


'KeepIsForm bevat IsForm in een bijzonder geval waarbij ladingen zijn betrokken


KeepIsForm = 0


'Letter spoort "e-" op in een woord


Letter = 0


'SRowNum is het rijnummer van de eerste tabelcel


'Wordt gebruikt om samengevoegde rijen cellen op te sporen


SRowNum = 0


'Dot is het aantal punten dat een CF nodig heeft voor de opmaak


Dot = 0








'Controle op 'no selection'; wordt getoond als de lengte = 1.


'Een (zinloze) 1-letter selectie wordt hierdoor ook opgespoord


'Als sel$=tablestring$ dan bevindt de cursor zich in een tabel, zonder een selectie


If Len(sel$) = 1 Or sel$ = TableString$ Then


    w = MsgBox(NoText$, 48)


    GoTo EndMainSub


End If





'Vind de start kolom en rij, en het totaal aantal kolommen


EndCol = Selection.Information(wdEndOfRangeColumnNumber)


StartCol = Selection.Information(wdStartOfRangeColumnNumber)


TotalCol = Selection.Information(wdMaximumNumberOfColumns)


StartRow = Selection.Information(wdStartOfRangeRowNumber)


EndRow = Selection.Information(wdEndOfRangeRowNumber)


totalrow = EndRow - StartRow + 1








If Asc(Mid$(sel$, 1, 1)) = 13 Then


    w = MsgBox(NoText$, 48)


    End


End If








'Als er kolommen zijn geselecteerd dan stoppen om foute opmaak te voorkomen


'Als het aantal geselecteerde kolommen <totaal aantal kolommen EN >1 geselecteerde rij:


If EndCol - StartCol + 1 < TotalCol And totalrow > 1 Then


    msgtext$ = "ChemFormula kan geen kolommen in een tabel opmaken"


    msgtext$ = msgtext$ + " tenzij de hele tabel is geselecteerd."


    d = MsgBox(msgtext$, 16)


    End


End If





'Ga na of selectie celgrenzen bevat


If InStr(sel$, TableString$) = 0 Then


    Boundary = 0


Else


    Boundary = 1


End If





If TotalCol = EndCol - StartCol + 1 And Boundary = 1 Then


    'Alle kolommen zijn geselcteerd en sommige celgrenzen zijn geselecteerd


    '(geen selectie binnen een cel)


    'Zorg dat alle CRs geselecteerd zijn, dan kan tabel correct worden opgemaakt


    'Als het nog niet gebeurd is, selecteer CRs


        If Asc(Mid$(sel$, Len(sel$) - 2, 1)) <> 7 Then


            Selection.MoveRight , 2, 1


        End If


    'Re-set Sel$ met ingesloten CRs


    sel$ = Selection


End If








'Er is een geldige selectie gemaakt.


'DotArraySize is de grootte van de matrix waarin punt informatie wordt bewaard


DotArraySize = Len(sel$)





'Definieer matrices voor verhoogde punt informatie


'DotArraySize is de pre-vastgestelde grootte van de matrix


ReDim hydratedot(DotArraySize)


ReDim posdots(DotArraySize)





StartSelection:





'Is het laatste teken een CR?


'Een poging om een CR teken (13) op te maken veroorzaakt een fout in W6


If Asc(Mid$(sel$, Len(sel$), 1)) = 13 Then


    'Stop macro met opmaak van laatste teken


    EndPos = Len(sel$) - 1


Else


    'Laatst teken is ok


    EndPos = Len(sel$)


End If











'Ga naar begin van de selectie


Selection.MoveLeft , , 0








For CurPos = 1 To EndPos





'Selecteer teken


Call CRight(1)





'Vind lengte van selectie (velden hebben lengte>1)


If Len(Selection) > 1 And Asc(Selection) <> 13 Then


    'Dit is een veld


    '(Als Asc(Selection) = 13 dan is het een tabel)


    'Verhoog CurPos overeenkomstig


    CurPos = CurPos + Len(Selection) - 1


End If








If CurPos > EndPos Then GoTo EndMainSub





'Vind ASCII van huidig teken


AscCur = Asc(Mid$(sel$, CurPos, 1))





'Vind ASCII van volgend teken


If CurPos < EndPos Then


    AscNext = Asc(Mid$(sel$, CurPos + 1, 1))


Else


    AscNext = 0


End If





'Vlag op als teken een hoofdletter alpha is


If AscCur >= 65 And AscCur <= 90 Then IsForm = 1


'Vlag op als e en volgend teken is "-" (wordt "e-")


'Behalve als voorgaand teken een letter is(dan is e deel van een woord)


If AscCur = 101 And (AscNext = 45 Or AscNext = 30) And Letter <> 1 _


Then IsForm = 1





'Subscript getallen waar nodig


If IsForm = 1 Then


    'Is teken een getal


    If AscCur >= 48 And AscCur <= 57 Then


        Selection.Font.Superscript = 0


        Selection.Font.Subscript = 1


    End If


Else


    'Teken is geen getal


    Selection.Font.Subscript = 0


End If











'Sla positie van hydraat-punten op


'Als vlag op, en teken is een punt (gewoon of verhoogd)


If IsForm = 1 And (AscCur = 46 Or AscCur = 183) Then


    'Vervang N door AscNext (vermijd lange regel)


    N = AscNext


    'Als volgend teken is hoofdletter alpha of numeriek


    If (N >= 65 And N <= 90) Or (N >= 48 And N <= 57) Then


        Dot = Dot + 1


        'Is het een stop of verhoogde punt


        t = MsgBox(DotQ$, 36)


        If t = 6 Then


            'Het is een verhoogde punt


            hydratedot(Dot) = 1


        Else


            'Geen verhoogde punt


            hydratedot(Dot) = 0


        End If


        'Sla positie van punt op


        posdots(Dot) = Selection.start


    End If


End If








'Geen + of -, of vlag is neer, dan deze paragraaf over slaan


If (AscCur <> 43 And AscCur <> 45 And AscCur <> 30) Or IsForm <> 1 _


Then GoTo AfterNoSuper


    


'Als volgend teken CR is of het eind van de selectie dan moet het lading zijn


If AscNext = 13 Or CurPos = EndPos Then GoTo AfterChargeQ


    


'Is het huidige teken een lading?


If AscCur <> 30 Then


    'Not a non-breaking hyphen


    w = MsgBox("Moet de " + Mid$(sel$, CurPos, 1) + SuperQ$, 36)


Else


    'Non-breaking hyphen


    'Cannot display a non-breaking hyphen in a msgbox, so show "-"


    w = MsgBox("Moet de -" + SuperQ$, 36)


End If


'Zo niet, de paragraaf over slaan


If w <> 6 Then GoTo NoSuper


AfterChargeQ:





'Is volgend teken een getal?


'Backwards charges


If AscNext >= 48 And AscNext <= 57 Then


    char = 0


    'Superscript het ladingsteken


    Selection.Font.Superscript = 1


    Call CRight(0)


    Call CRight(1)


    CurPos = CurPos + 1


    p = Asc(Mid$(sel$, CurPos, 1))


    'Ga door zolang het volgend teken een cijfer is


    While p >= 48 And p <= 57


        'Superscript het cijfer


        Selection.Font.Superscript = 1


        char = char + 1


        'Voorbij het eind van de selectie


        If CurPos + char > EndPos Then p = 0


        If p <> 0 Then


            'Vind ASCII van volgend teken


            p = Asc(Mid$(sel$, CurPos + char, 1))


        End If


        'Schuif op naar volgend teken


        Call CRight(0)


        Call CRight(1)


    Wend


    'Verhoog CurPos, omdat de selectie doorlopen is


    CurPos = CurPos + char - 1


    'Ga een teken terug (we gingen te ver)


    Selection.MoveLeft , , 0


    Selection.MoveLeft , , 1


    'Stop EOF fouten


    If CurPos > EndPos Then CurPos = EndPos


    'Houd vlag op hier


    KeepIsForm = 1


    'Superscripting is gebeurd, verlaat superscript gebied


    GoTo AfterNoSuper


End If





f = Asc(Mid$(sel$, CurPos - 1, 1))





'Is voorgaand teken een cijfer


If f < 48 Or f > 57 Then


    'Geen cijfer


    Selection.Font.Superscript = 1


Else


    'Voorgaand teken is cijfer


    'Selecteer het voorgaand teken ook


    Call CRight(0)


    Selection.MoveLeft , , 1


    Selection.MoveLeft , , 1


    w = MsgBox("Should the " + Chr$(f) + SuperQ$, 36)


    If w = 6 Then


        'Super dit teken en het voorafgaande


        Selection.Font.Superscript = 1


    Else


        'Super alleen huidig teken


        Call CRight(0)


        Selection.MoveLeft , , 1


        Selection.Font.Superscript = 1


    End If


End If


'IsForm moet zo blijven in deze situatie


KeepIsForm = 1


GoTo AfterNoSuper





'Huidig teken hoeft niet superscript te worden


NoSuper:


Selection.Font.Superscript = 0


AfterNoSuper:





'Haal vlag neer nij niet-chemische formule tekens


'<47 zijn symbolen en nietafdrukbare tekens


'40 en 41 zijn () die vlag -niet- moeten neer halen


If AscCur <= 47 And AscCur <> 41 And AscCur <> 40 Then IsForm = 0


'Zet IsForm weer aan als een lading zojuist is opgemaakt


If KeepIsForm = 1 And (AscCur = 43 Or AscCur = 45 Or AscCur = 30) Then


    IsForm = 1


    KeepIsForm = 0


End If


'58-64 zijn weer symbolen  (lals :;@ etc.)


If AscCur >= 58 And AscCur <= 64 Then IsForm = 0


'91-96 zijn meer zoals ^_


'93 is ] die de vlag op moeten laten (bij TM zaken)


If AscCur >= 91 And AscCur <= 96 And AscCur <> 93 Then IsForm = 0


'>123 zijn symbolen


If AscCur >= 123 Then IsForm = 0


'Als teken een letter is, sla dit teken op (voor <letter>e- opsporing)


If AscCur >= 97 And AscCur <= 122 Then Letter = 1 Else Letter = 0





Call CRight(0)





Next CurPos








'Sla eind van selectie op, voor grote schoonmaak later


SelEnd = Selection.start








FormatDots:


'Maak punten op


For h = 1 To Dot


'Selecteer de betrokken punt


Selection.start = posdots(h)


Selection.End = posdots(h) + 1


Selection.MoveLeft


Selection.start = posdots(h)


Selection.End = posdots(h) + 1


If hydratedot(h) = 1 Then


    'Verhoogde punt


    Selection = Chr$(183)


Else


    'Punt


    Selection = Chr$(46)


End If


Next h








'Schoonmaak


'Selecteer laatste teken


Selection.start = SelEnd - 1


Selection.End = SelEnd


'Ga door, niet is geselecteerd


Selection.MoveRight


'Zet sub- en superscripting uit


Selection.Font.Subscript = 0


Selection.Font.Superscript = 0





EndMainSub:


End Sub





Sub CRight(Extend)


If Extend = 0 Then


    'Ga naar rechts zonder selectie uit te breiden


    Selection.MoveRight


Else


    'Breid de selectie uit met 1 naar rechts


    Selection.MoveRight , , 1


    'Als selectie in tabel zit


    If Selection.Information(wdWithInTable) = True Then


        CurRowNum = Selection.Information(wdStartOfRangeRowNumber)


        'Niet de eerste rij in de selectie


        If SRowNum <> 0 Then


            If CurRowNum <> SRowNum Then


                'Selectie is niet op te maken


            'Ga naar de geode plaats , zodat gebruiker verder kan gaan


                Selection.MoveLeft , 2, 0


                'Toon bericht


                e = MsgBox(MRow$, 48)


                'Stop uitvoering


                End


            End If


        End If


        'Als eerste rij in selectie zit, sla rij nummer op


        If SRowNum = 0 Then SRowNum = CurRowNum


    End If


    If Len(Selection) > 1 Then


        'Veld of hele tabelcel is geselecteerd


        'Als selectie een table is


        If Selection.Information(wdWithInTable) = True Then


            IsForm = 0


            'Ga weer 1 teken naar rechts (naar de volgende cel)


            Selection.MoveRight


            Selection.MoveRight , , 1


            'Verhoog CurPos voorbij de CRs


            CurPos = CurPos + 2


            If Asc(Selection) = 13 Then


                'Einde van tabel rij


                CurPos = CurPos + 1


                'Re-set start rij nummer, meerdere rijen is toegestaan


                SRowNum = 0


            End If


        End If


    End If


End If


End Sub








